
概 要
　高機能高性能の Analog-VLSIを短期間で効率良く、しかも一定の確実性をもって開発するこ

とができるような礎を構築すべく、公開の IPとして Analog-VLSIのビルディングブロックを提供
します。第 4 章では、増幅器回路、抵抗回路等において用いられている定電流源に基準電圧を供給
するための IPを提示致します。

1 バイアス回路

バイアス回路とは、増幅要素、抵抗回路要素等において用いられている定電流源のための基準電圧

を発生する回路をいいます。
∗ikeda.hirokazu@jaxa.jp

増幅要素、抵抗回路等においては、一定の関係を有する複数の定電流源を有しています。これらの

定電流源は、一般的には、カスコードトランジスタを伴ったソース接地のトランジスタから構成され

ており、そのゲートに一定の電圧を印加することにより定電流源として動作するようになっています。

しかし、ドレイン電流は、ゲート電圧によって敏感に変動するため、ゲート電圧を外部から直接印

加することにより定電流源の電流値を制御することは困難です。また、プロセス変動により当該電圧

値は微妙に変動します。

そこで、外部から基準となる電流を印加し、これによって間接的に所要のゲート電圧を発生する回

路を設けることが便宜です。

このような回路として、バイアス回路を用意しました。バイアス回路は、VDD側の定電流源（source）
である PMOSトランジスタのゲートに印加する電圧 VHと、VSS側の定電流源 (sink)である NMOS
トランジスタのゲートに印加する電圧 VL、とカスコードトランジスタのゲートに印加する電圧 (VM)
とを供給するようになっています。VMは、GNDと同一の電位ですが、GNDを経由した、信号チェ
インの上流下流間の干渉を防止するため、及びゲートの静電気破壊を防止するため 1 kΩの抵抗が直列
に挿入されています。

また、表 1には、本 IPにおけるバイアス回路の名称と、それらの機能の概略をまとめました。

表 1: バイアス回路 IP

回路名称 用途 具体的適用

BIAS 1:1/10の基準電圧発生回路　 増幅器のバイアス用　

BIASP 1:1/10の基準電圧発生回路 　 ロジックインターフェース用

BIAS1 1:1/10の基準電圧発生回路　 バイアス電流を電圧に変換　

BIAS2 1:1/2の基準電流発生回路 　 バイアス電流の減衰調整　

BIAS4 1:1/4の基準電流発生回路 　 バイアス電流の減衰調整　

BIAS8 1:1/8の基準電流発生回路　 バイアス電流の減衰調整

BIASF 1:10/D(0:3)の基準電流発生回路 バイアス電流の減衰調整

BIASGEN BIAS回路に電流分配機能を付加　 バイアス電流の分配

図 1には、これらのバイアス回路のシンボルの一覧を掲げました。

BI
AS

1

IIN

VH

VL

BI
AS

2IOUT

IIN

BI
AS

4

IIN

IOUT

BI
AS

8

IIN

IOUT

BIASF
D0 D1 D2 D3

IIN IOUT

2

3

4

5

BI
AS

-G
EN

IIN 

VH1

VL1

BI
AS

P

DG
IIN 

VH

VL

VM

BI
AS

IIN 

VH

VL

VM

図 1: バイアス回路のシンボル一覧

以下、各回路の回路図面を掲げます。

図 2: BIAS回路

図 3: BIASP回路

図 4: BIAS1回路

図 5: BIAS2回路

図 6: BIAS4回路

図 7: BIAS8回路

図 8: BIASF回路

図 9: BIASGEN回路

以下に上記 IPのネットリストを掲げました。

.SUBCKT BIAS IIN VH VL VM Gnd

C1 IIN VH 1pF

C2 N2 VL 1pF

M4 IIN IIN N2 VSS nch L=1u W=3u M=10

M3 VH IIN VL VSS nch L=1u W=3u M=10

M2 VL VL VSS VSS nch L=3u W=3u M=10

M1 N2 VL VSS VSS nch L=3u W=3u M=10

M5 VH VH VDD VDD pch L=3u W=3.u M=10

R3 Gnd VM 1K TC=0.0, 0.0

.ENDS

.SUBCKT BIASP DG IIN VH VL VM

C1 IIN VH 1pF

C2 N8 VL 1pF

M4 IIN IIN N8 VSS nch L=1u W=3u M=10

M3 VH IIN VL VSS nch L=1u W=3u M=10

M2 VL VL VSS1 VSS nch L=3u W=3u M=10

M1 N8 VL VSS1 VSS nch L=3u W=3u M=10

M5 VH VH VDD1 VDD1 pch L=3u W=3.u M=10

R3 DG VM 1K TC=0.0, 0.0

.ENDS

.SUBCKT BIAS1 IIN VH VL

M1 VL VL VSS VSS nch L=3u W=3u M=10

M2 VH IIN VL VSS nch L=1u W=3u M=10

M3 VH VH VDD VDD pch L=3u W=3.u M=10

C4 VH IIN 1pF

C5 VL N3 1pF

M6 IIN IIN N3 VSS nch L=1u W=3u M=10

M7 N3 VL VSS VSS nch L=3u W=3u M=10

.ENDS

.SUBCKT BIAS2 IIN IOUT

C1 N2 N5 1pF

C2 IIN N3 1pF

M3 N2 N5 VSS VSS nch L=3u W=3u M=10

M4 IIN IIN N2 VSS nch L=1u W=3u M=10

M5 N5 N5 VSS VSS nch L=3u W=3u M=10

M6 N3 IIN N5 VSS nch L=1u W=3u M=10

M7 IOUT N3 VDD VDD pch L=3u W=3.u M=10

M8 N3 N3 VDD VDD pch L=3u W=3.u M=20

.ENDS

.SUBCKT BIAS4 IIN IOUT

C1 N10 N12 1pF

C2 IIN N3 1pF

M3 IIN IIN N10 VSS nch L=1u W=3u M=10

M4 N10 N12 VSS VSS nch L=3u W=3u M=10

M5 N12 N12 VSS VSS nch L=3u W=3u M=10

M6 N3 IIN N12 VSS nch L=1u W=3u M=10

M7 IOUT N3 VDD VDD pch L=3u W=3.u M=10

M8 N3 N3 VDD VDD pch L=3u W=3.u M=40

.ENDS

.SUBCKT BIAS8 IIN IOUT

M1 M1D M1D VSS VSS nch L=3u W=3u M=10

M2 N1 IIN M1D VSS nch L=1u W=3u M=10

M3 N1 N1 VDD VDD pch L=3u W=3.u M=80

M4 IOUT N1 VDD VDD pch L=3u W=3.u M=10

C5 N2 M1D 1pF

C6 IIN N1 1pF

M7 N2 M1D VSS VSS nch L=3u W=3u M=10

M8 IIN IIN N2 VSS nch L=1u W=3u M=10

.ENDS

.SUBCKT BIASF D0 D1 D2 D3 IIN IOUT Gnd

XBIAS_1 IIN N42 N34 N35 Gnd BIAS

C1 N38 N34 4pF

M2 N38 D0 N33 VSS nch L=1u W=3u M=1

M3 N33 N34 VSS VSS nch L=3u W=3u M=1

M4 N38 D1 N40 VSS nch L=1u W=3u M=2

M5 N40 N34 VSS VSS nch L=3u W=3u M=2

M6 N38 D2 N41 VSS nch L=1u W=3u M=4

M7 N41 N34 VSS VSS nch L=3u W=3u M=4

M8 N38 D3 N37 VSS nch L=1u W=3u M=8

M9 N37 N34 VSS VSS nch L=3u W=3u M=8

M10 N39 N35 N38 VSS nch L=1u W=3u M=10

M11 IOUT N39 VDD VDD pch L=3u W=3u M=10

M12 N39 N39 VDD VDD pch L=3u W=3u M=10

.ENDS

.SUBCKT BIASGEN IIN IOUT2 IOUT3 IOUT4 IOUT5 VH1 VL1

C1 N1 VL1 1pF

C2 IIN VH1 1pF

M3 N1 VL1 VSS VSS nch L=3u W=3u M=10

M4 VL1 VL1 VSS VSS nch L=3u W=3u M=10

M5 VH1 IIN VL1 VSS nch L=1u W=3u M=10

M6 IIN IIN N1 VSS nch L=1u W=3u M=10

M7 VH1 VH1 VDD VDD pch L=3u W=3.u M=10

M8 IOUT2 VH1 VDD VDD pch L=3u W=3.u M=10

M9 IOUT3 VH1 VDD VDD pch L=3u W=3.u M=10

M10 IOUT4 VH1 VDD VDD pch L=3u W=3.u M=10

M11 IOUT5 VH1 VDD VDD pch L=3u W=3.u M=10

.ENDS

A 来歴

• 定電流源の IPでは、ソース接地されているトランジスタの L値として 1 µm を用いていました。

しかし、0.25 µmのプロセスにおいてそのまま適用するとバイアス源の雑音が信号系統の信号

対雑音比に顕著に影響することが分かりました。同じサイズのトランジスタであれば toxが薄く

なった分 gm が増加するからです。

そこで、L値を 3 µmとするとともに、バイアス回路に低周波通過特性を持たせるように改修し

ました。これにともなって対応する増幅器の定電流源用のトランジスタの L値も 3 µmとする必

要があります。

このような対策によって、0.25 µm のプロセスにおける低雑音化が達成されたばかりでなく、

0.35 µmのプロセスにおいても有意な低雑音化が図られることが分かりました (H160817)。

以上

